A

AUTOEBALUAZIO ARIKETAK

Proposatutako Ariketa A-1

a) Definitu ondoko bi funtzioak: f(x,y) = sin(x)sin(y) - 0,5 eta g(x,y) = cos(x)cos(y) - 0,5.
b) Egin f(x,y) =0 eta g(x,y) =0 kurben adierazpen grafikoa ardatz berdinak erabiliz, bakoitzari kolore ezberdinak egokitu eta grafikoaren atzealdea ere koloreztatu.

Soluzioa A-1

a) Funtzioen definizioa

```
f[x_, y_] = Sin[x] * Sin[y] - 0.5;
g[x_, y_] = Cos[x] * Cos[y] - 0.5;
```

b) Funtzioen adierazpen grafikoa

```
ContourPlot[{f[x, y] == 0, g[x, y] == 0}, {x, -2 π, 2 π}, {y, -2 π, 2 π},
 ContourStyle → {{Thickness[0.01], Yellow}, {Thickness[0.01], Green}},
 Axes → True, AxesLabel → {"OX", "OY"}, Background → LightBlue]
```



```
ContourPlot[{f[x, y]}, {x, -2 π, 2 π}, {y, -2 π, 2 π},
 ContourStyle → {{Thickness[0.01], Yellow}, {Thickness[0.01], Green}},
 Axes → True, AxesLabel → {"OX", "OY"}, Background → LightBlue]
```



▼ Proposatutako Ariketa A-2

Marraztu errosazeoen ondoko familia:



▼ Soluzioa A-2

```
g1 = PolarPlot[Sin[4 θ], {θ, 0, 2 Pi}, AxesLabel → {x, y}, Mesh → 15,
    MeshFunctions → {#1 &}, MeshStyle → Red, PlotLabel → r == Sin[4 θ]];
g2 = PolarPlot[Sin[6 θ], {θ, 0, 2 Pi}, AxesLabel → {x, y}, PlotStyle → Thick,
    Mesh → 15, MeshShading → {Orange, Green}, PlotLabel → r == Sin[6 θ]];
g3 = PolarPlot[Cos[6 θ], {θ, 0, 2 Pi}, AxesLabel → {x, y}, PlotStyle → Thick,
    ColorFunction → Function[{x, y, θ}, Hue[θ / (4 Pi)]],
    ColorFunctionScaling → False, PlotLabel → r == Sin[6 θ]];
g4 = PolarPlot[{4 + 3 * Sin[12 * (t - 0.1)], 4 + 3 * Cos[12 * t], 1, 7}, {t, 0, 2 π},
   ColorFunction → "DarkRainbow", PlotStyle → Directive[Red, Thick]];
GraphicsGrid[{{g1, g2}, {g3, g4}}]
```



▼ Proposatutako Ariketa A-3

Marraztu kardioideen ondoko familia:



▼ Soluzioa A-3

1. Kardioidea

```
kardioide1[t_, a_] = a (1 + Cos[t]);
```

2. Kardioidea

```
kardioide2[t_, a_] = a (1 - Cos[t]);
```

3. Kardioidea

```
kardioide3[t_, a_] = a (1 + Sin[t]);
```

4. Kardioidea

```
kardioide4[t_, a_] = a (1 - Sin[t]);
```

Kardioide familia

```
c1 = PolarPlot[Evaluate[Table[kardioide1[t, a], {a, 1, 10, 2}]], {t, 0, 2 π}, PlotStyle →
    Table[RGBColor[0, 0.5, i * 0.1], {i, 1, 10, 2}], PlotLabel → r == a (1 + Cos[t])];
c2 = PolarPlot[Evaluate[Table[kardioide2[t, a], {a, 1, 10, 2}]], {t, 0, 2 π},
  PlotStyle → Table[RGBColor[i * 0.1, 0, 0.5], {i, 1, 10, 2}],
  PlotLabel → r == a (1 - Cos[t])];
c3 = PolarPlot[Evaluate[Table[kardioide3[t, a], {a, 1, 10, 2}]], {t, 0, 2 π}, PlotStyle →
   Table[RGBColor[0.2, 0, i * 0.1], {i, 1, 10, 2}], PlotLabel → r == a (1 + Sin[t])];
c4 = PolarPlot[Evaluate[Table[kardioide4[t, a], {a, 1, 10, 2}]], {t, 0, 2 π},
  PlotStyle → Table[RGBColor[0.5, i * 0.1, 0], {i, 1, 10, 2}],
  PlotLabel → r == a (1 - Sin[t])]; GraphicsGrid[{{c1, c2}, {c3, c4}}]
```



▼ Proposatutako Ariketa A-4

Marraztu lemniskaten familia:



▼ Soluzioa A-4

```
eklemn = (x^4) == a^2 (x^2 - y^2)
```

$x^4 == a^2 (x^2 - y^2)$

```
eklemn = (x^4) == a^2 (x^2 - y^2) /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$a^2 \text{Cos}[2 t]\, r[t] == \text{Cos}[t]^4\, r[t]^3$

```
lemniskatak[t_, a_] = a (Cos[2 * t]) ^ (1 / 2) / Cos[t] ^ 2
```

$a\sqrt{\text{Cos}[2\,t]}\ \text{Sec}[t]^2$

```
PolarPlot[Evaluate[Table[lemniskatak[t, a], {a, 1, 9, 2}]], {t, 0, 2 π},
 PlotLabel → Style["GERONO LEMNISKATA FAMILIA", Bold, 14], Background → GrayLevel[0.8]]
```



▼ Proposatutako Ariketa A-5

Edozein bi funtzio emanda, definitu edozein puntutako funtzioarekiko zuzen ukitzaileak. Marraztu funtzioak eta zuzen ukitzaileak puntuaren ingurune batean.

▼ Soluzioa A-5

```
f[x_] = 2 * x / (x^2 + 1);
g[x_] = Cos[x];
x0 = 1 / 2;
```

```
tangentef[x_] = f[x0] + f '[x0] * (x - x0);
tangenteg[x_] = g[x0] + g '[x0] * (x - x0);
```

```
Plot[{f[x], tangentef[x], g[x], tangenteg[x]}, {x, -1, 2},
 PlotStyle → {{Blue, Thickness[0.01]}, {Orange, Thickness[0.005], Dashing[0.02]},
   {Green, Thickness[0.01]}, {Purple, Thickness[0.005], Dashing[0.02]}}]
```



▼ Proposatutako Ariketa A-6

Egin ondoko grafikoaren animazioa:



▼ Soluzioa A-6

```
ek = (x - a)^2 + (y - b)^2 == c^2
```

$(-a + x)^2 + (-b + y)^2 == c^2$

```
ek3 = ek /. {a → 0, b → 0}
```

$x^2 + y^2 == c^2$

```
polar3 = ek3 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$c^2 == r[t]^2$

```
Solve[polar3, r[t]]
```

{{r[t] → -c}, {r[t] → c}}

```
zirk3[t_, c_] = c;
```

```
Manipulate[PolarPlot[{zirk3[t, 4], zirk3[t, c]}, {t, 0, 2 * π}, PlotStyle →
   {{RGBColor[0.5, 1, 0], Thickness[0.015]}, RGBColor[0.5, 0, 1]}], {c, 1, 4, 0.2}]
```



▾ Proposatutako Ariketa A-7

Egin ondoko grafikoaren animazioa:



▼ Soluzioa A-7

```
Manipulate[PolarPlot[Evaluate[Table[zirk3[t, 4 - p], {p, 0, c, 0.2}]], {t, 0, 2 * π}],
 {c, 0.2, 4, 0.2}]
```



▼ Proposatutako Ariketa A-8

Egin ondoko grafikoaren animazioa:



▼ Soluzioa A-8

```
ek1 = ek /. {a → 0, c → b}
```

$x^2 + (-b + y)^2 == b^2$

```
polar1 = ek1 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$r[t]^2 == 2\,b\,r[t]\,Sin[t]$

```
Solve[polar1, r[t]]
```

```
{{r[t] → 0}, {r[t] → 2 b Sin[t]}}
```

```
zirk1[t_, b_] = 2 * b Sin[t];
Manipulate[PolarPlot[zirk1[t, 1], {t, 0, p}, PlotStyle → RGBColor[0.5, 0, 1]], {p, 0.1, π}]
```



▼ Proposatutako Ariketa A-9

Egin ondoko grafikoaren animazioa:



▼ Soluzioa A-9

```
ek2 = ek /. {b → 0, c → a}
```

$(-a+x)^2 + y^2 == a^2$

```
polar2 = ek2 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$2\,a\,\mathrm{Cos}[t]\,r[t] == r[t]^2$

$\mathbf{2\,a\,Cos[t]\,r[t] == r[t]^2}$

$2\,a\,\mathrm{Cos}[t]\,r[t] == r[t]^2$

```
Solve[polar2, r[t]]
```

{{r[t] → 0}, {r[t] → 2 a Cos[t]}}

```
zirk2[t_, a_] = 2 * a Cos[t];

Manipulate[PolarPlot[zirk2[t, 1], {t, 0, p}, PlotStyle → RGBColor[0.5, 0, 1]], {p, 0.1, π}]
```



▼ Proposatutako Ariketa A-10

Egin ondoko grafikoaren animazioa:



▼ Soluzioa A-10

```
Manipulate[PolarPlot[Evaluate[Table[circulo2[t, 4 - p], {p, 0, c, 0.2}]], {t, 0, 2 * π}],
 {c, 0.2, 4, 0.2}]
```



▼ Proposatutako Ariketa A-11

Egin ondoko grafikoaren animazioa:



▼ Soluzioa A-11

```
Manipulate[
 PolarPlot[Evaluate[Table[zirk1[t, 4 - p], {p, 0, c, 0.2}]], {t, 0, 2 * π}], {c, 0, 4, 0.2}]
```



▼ Proposatutako Ariketa A-12

a) Aztertu $f(x,y)=\frac{x^2-y^2}{x^2+y^2}$ funtzioaren errepikatutako limiteen eta limite erradialen existentzia.

b) Aztertu $f(x,y)=\frac{x\,y}{x^2+y^4}$ funtzioaren errepikatutako limiteen eta limite erradialen existentzia.

▼ Soluzioa A-12

a) Atala

```
f[x_, y_] = (x^2 - y^2) / (x^2 + y^2)
```

$$\frac{x^2 - y^2}{x^2 + y^2}$$

Errepikatutako limiteak

```
l1 = Limit[Limit[f[x, y], x -> 0], y -> 0]
```

-1

```
l2 = Limit[Limit[f[x, y], y -> 0], x -> 0]
```

1

Norabide limiteak

```
Limit[f[x, m * x], x -> 0]
```

$$\left\{\left\{\text{Limit}\left[\frac{x^2 - x^2\left(y'[x] \to \frac{-x^2+y[x]^2}{2\,x\,y[x]}\right)^2}{x^2 + x^2\left(y'[x] \to \frac{-x^2+y[x]^2}{2\,x\,y[x]}\right)^2},\ x \to 0\right]\right\}\right\}$$

Limite erradialak ez dira existitzen

```
Plot3D[f[x, y], {x, -4, 4}, {y, -4, 4}, Mesh → False]
```



```
ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1}]
```



b) Atala

```
f[x_, y_] = (x * y) / (x^2 + y^4)
```

$$\frac{x\,y}{x^2 + y^4}$$

Errepikatutako limiteak

Ez da existitzen f2[y] funtzio marjinala x→0 denean

```
l1 = Limit[Limit[f[x, y], x -> 0], y -> 0]
```

0

```
l2 = Limit[Limit[f[x, y], y -> 0], x -> 0]
```

0

Norabide limiteak

```
Limit[f[x, m * x], x -> 0]
```

$$\left\{\left\{\text{Limit}\left[\frac{x^2 \left(y'[x] \to \frac{-x^2+y[x]^2}{2\,x\,y[x]}\right)}{x^2 + x^4 \left(y'[x] \to \frac{-x^2+y[x]^2}{2\,x\,y[x]}\right)^4},\ x \to 0\right]\right\}\right\}$$

Limite erradialak ez dira existitzen

```
Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}, Mesh → False]
```



```
ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1}]
```



▾ Proposatutako Ariketa A-13

Koordenatu jatorrian OY ardatzarekiko ukitzaileak diren zirkulu familia emanik $x^2+ y^2=cx$,

a) Lortu berari lotutako Ekuazio Diferentziala (E.D.) eta ebatzi.

b) Lortu soluzioen familia bat eta marraztu.

c) Lortu ibilbide ortogonalen E.D eta ebatzi.

d) Lortu soluzioen familia bat eta marraztu.

e) Marraztu bi kurben familiak eta kurba familia bakoitzari lotutako eremu bektorialak ibilbide ortogonalekin batera.

▼ Soluzioa A-13

a) Atala

■ **Kurba familiaren ekuazioa $x^2 + y^2 = c * x$**

```
ek = x^2 + y[x]^2 == c * x
```

$x^2 + y[x]^2 == c\,x$

```
con = Solve[ek, c]
```

$\left\{\left\{c \to \frac{x^2 + y[x]^2}{x}\right\}\right\}$

```
ed = D[ek, x] /. c → con[[1, 1, 2]]
```

$2\,x + 2\,y[x]\,y'[x] == \frac{x^2 + y[x]^2}{x}$

b) Atala

■ **Kurba familiari lotutako Ekuazio Diferentzial Arrunta ebatziko dugu**

```
Si = DSolve[ed, y[x], x]
```

$\left\{\left\{y[x] \to -\sqrt{-x^2 + x\,C[1]}\right\}, \left\{y[x] \to \sqrt{-x^2 + x\,C[1]}\right\}\right\}$

```
s1[x_, c_] = Si[[1, 1, 2]] /. C[1] → c / 2
s2[x_, c_] = Si[[2, 1, 2]] /. C[1] → c / 2
```

$-\sqrt{\frac{c\,x}{2} - x^2}$

$\sqrt{\frac{c\,x}{2} - x^2}$

■ **Soluzio familia bat honako zerrenda da**

```
solti = Table[{s1[x, c], s2[x, c]}, {c, -20, 20., 2}];

listasolti = Flatten[solti, 2];

famsolti = Plot[Evaluate[listasolti], {x, -10, 10}, PlotStyle →
    {{Orange, Thickness[0.004]}, {Orange, Thickness[0.004]}}, AspectRatio → Automatic]
```



■ **(x,y) puntu bakoitzean, kurbarekiko ukitzailearen bektore zuzentzailea (1,m) da, non**

```
m = Solve[ed, y'[x]]
```

$\left\{\left\{y'[x] \to \frac{-x^2 + y[x]^2}{2\,x\,y[x]}\right\}\right\}$

c) Atala

- **Ibilbide ortogonalen E. D.**

```
edto = y'[x] == -1 / m[[1, 1, 2]]
```

$$y'[x] == -\frac{2\,x\,y[x]}{-x^2 + y[x]^2}$$

```
So = DSolve[edto, y[x], x]
```

$$\left\{\left\{y[x] \to \frac{1}{2}\left(e^{C[1]} - \sqrt{e^{2\,C[1]} - 4\,x^2}\right)\right\}, \left\{y[x] \to \frac{1}{2}\left(e^{C[1]} + \sqrt{e^{2\,C[1]} - 4\,x^2}\right)\right\}\right\}$$

```
so1[x_, c_] = So[[1, 1, 2]] /. {e^C[1] → c, e^(2*C[1]) → c^2}
so2[x_, c_] = So[[2, 1, 2]] /. {e^C[1] → c, e^(2*C[1]) → c^2}
```

$$\frac{1}{2}\left(c - \sqrt{c^2 - 4\,x^2}\right)$$

$$\frac{1}{2}\left(c + \sqrt{c^2 - 4\,x^2}\right)$$

d) Atala

- **Soluzio familia bat honako zerrenda da**

```
solto = Table[{so1[x, c], so2[x, c]}, {c, -10, 10., 1}];

listasolto = Flatten[solto, 2];

famsolto = Plot[Evaluate[listasolto], {x, -5, 5}, PlotStyle →
   {{Blue, Thickness[0.004]}, {Blue, Thickness[0.004]}}, AspectRatio → Automatic]
```




```
s[x_, c_] = So[[1, 1, 2]] /. C[1] → c
```

$$\frac{1}{2}\left(e^{c} - \sqrt{e^{2\,c} - 4\,x^2}\right)$$

e) Atala

- **Kurba familia eta berarekiko ibilbide ortogonalak marraztuko ditugu**

```
j = Show[{famsolto, famsolti},
  AspectRatio → Automatic, PlotRange → {{-10, 10}, {-10, 10}}]
```



- **Kurba familiari eta ibilbide ortogonalei lotutako eremu bektorialak marraztuko ditugu**

```
m
```

$$\left\{\left\{y'[x] \to \frac{-x^2 + y[x]^2}{2\,x\,y[x]}\right\}\right\}$$

```
k = VectorPlot[{{2 * x * y, y^2 - x^2}, {-y^2 + x^2, 2 * x * y}},
  {x, -10, 10}, {y, -10, 10}, VectorPoints → 20, VectorScale → Small,
  StreamScale → Full, VectorStyle → {Purple, Blue}]
```




```
Show[j, k]
```